Supplementary Materials: Joint Modeling of Multiple Crimes: A Bayesian Spatial Approach

Hongqiang Liu and Xinyan Zhu

WinBUGS code for the univariate model

```
model {
        for (i in 1:Nareas) {
            for (k in 1:Ndiseases) {
                Y[i, k] ~ dpois(mu[i, k])
                log(mu[i, k]) <- log(E[i, k]) + alpha[k] + beta[k, 1] * POPDENSITY[i] +
beta[k, 2] * UNEMPRATE[i] + beta[k, 3] * PHE[i] + beta[k, 4] * PYM[i] + beta[k, 5] * POL[i] +
beta[k, 6] * BARDENSITY[i] + s[k, i] + u[i, k]
               u[i, k] ~ dnorm(0, tau.u[k])
            }

        # area specific relative risk for burglaries
        RR1[i] <- exp(alpha[1] + beta[1, 1] * POPDENSITY[i] + beta[1, 2] * UNEMPRATE[i] +
  beta[1, 3] * PHE[i] + beta[1, 4] * PYM[i] + beta[1, 5] * POL[i] + beta[1, 6] * BARDENSITY[i]+
  s[1, i] + u[i, 1])
        # area specific relative risk for non-motor vehicle thefts
        RR2[i] <- exp(alpha[2] + beta[2, 1] * POPDENSITY[i] + beta[2, 2] * UNEMPRATE[i] +
  beta[2, 3] * PHE[i] + beta[2, 4] * PYM[i] + beta[2, 5] * POL[i] + beta[2, 6] * BARDENSITY[i] +
  s[2, i] + u[i, 2])
    }

    for (k in 1:Ndiseases) {s[k, 1:Nareas] ~ car.normal(adj[], weights[], num[], tau.s[k]) }
    for (k in 1:Ndiseases) {
             tau.s[k] ~ dgamma(0.5, 0.0005)
             tau.u[k] ~ dgamma(0.5, 0.0005)

             sigma2.s[k] <- 1/tau.s[k] # variance
             sigma2.u[k] <- 1/tau.u[k] # variance
        }

        for (i in 1:sumNumNeigh) {weights[i] <- 1 }
            for (i in 1:6) {
            beta[1, i] ~ dnorm(0.0, 1.0E-4)
            beta[2, i] ~ dnorm(0.0, 1.0E-4)
            }
        for (k in 1:Ndiseases) {
            alpha[k] ~ dflat()
        }
}
```

WinBUGS code for the multivariate model

```
model {
        for (i in 1:Nareas) {
          for (k in 1:Ncrimes) {
            Y[i, k] ~ dpois(mu[i, k])
            log(mu[i, k]) <- log(E[i, k]) + alpha[k] +    beta[k, 1] * POPDENSITY[i] + beta[k, 2] *
  UNEMPRATE[i] + beta[k, 3] * PHE[i] + beta[k, 4] * PYM[i] + beta[k, 5] * POL[i] + beta[k, 6] *
  BARDENSITY[i] + s[k, i] + u[i, k]
        }
        # area specific relative risk for burglaries
        RR1[i] <- exp(alpha[1] + beta[1, 1] * POPDENSITY[i] + beta[1, 2] * UNEMPRATE[i] +
  beta[1, 3] * PHE[i] + beta[1, 4] * PYM[i] + beta[1, 5] * POL[i] + beta[1, 6] * BARDENSITY[i]+
  s[1, i] + u[i, 1])
        # area specific relative risk for non-motor vehicle thefts
        RR2[i] <- exp(alpha[2] + beta[2, 1] * POPDENSITY[i] + beta[2, 2] * UNEMPRATE[i] +
  beta[2, 3] * PHE[i] + beta[2, 4] * PYM[i] + beta[2, 5] * POL[i] + beta[2, 6] * BARDENSITY[i] +
  s[2, i] + u[i, 2])
  }

    s[1:Ncrimes, 1:Nareas] ~ mv.car(adj[], weights[], num[], omega[ , ])

    for (i in 1:sumNumNeigh) { weights[i] <- 1 }

    for (i in 1:6) {
        beta[1, i] ~ dnorm(0.0, 1.0E-4)
        beta[2, i] ~ dnorm(0.0, 1.0E-4)
    }

    for (i in 1:Nareas) {
        #Unstructured multivariate normal
        u[i, 1:Ncrimes] ~ dmnorm(zero[], tau[ , ])
    }

    for (k in 1:Ncrimes) {
        alpha[k] ~ dflat()
    }
    omega[1:Ncrimes, 1:Ncrimes] ~ dwish(R[ , ], Ncrimes)        # Precision matrix
    sigma2.s[1:Ncrimes, 1:Ncrimes] <- inverse(omega[ , ])       # Covariance matrix

    sigma.s[1] <- sqrt(sigma2.s[1, 1])
    sigma.s[2] <- sqrt(sigma2.s[2, 2])

    # within-area conditional correlation between spatial components
    corr.s <- sigma2.s[1, 2] / (sigma.s[1] * sigma.s[2])
```

```
    tau[1:Ncrimes, 1:Ncrimes] ~ dwish(Q[ , ], Ncrimes)     # Precision matrix
    sigma2.u[1:2, 1:2] <- inverse(tau[ , ])               # Covariance matrix

    sigma.u[1] <- sqrt(sigma2.u[1, 1])
    sigma.u[2] <- sqrt(sigma2.u[2, 2])

    # within-area correlation between unstructured components
    corr.u <- sigma2.u[1, 2] / (sigma.u[1] * sigma.u[2])
    # within-area conditional correlation between total random effects
    corr.sum <-  (sigma2.s[1, 2] + sigma2.u[1, 2]) / (sqrt(sigma2.s[1, 1] + sigma2.u[1, 1]) *
  sqrt(sigma2.s[2, 2] + sigma2.u[2, 2]))

    mu1 <-  mean(RR1[])
    mu2 <-  mean(RR2[])
    sd1 <- sd(RR1[])
    sd2 <- sd(RR2[])
    mu12 <- inprod(RR1[], RR2[])/Nareas

    #the correlation between the relative risks
    CRR12 <- (mu12 - mu1*mu2)/(sd1*sd2)
}
```